2.8 Complete Elliptic Integrals K and E

A. Purpose

These subprograms compute values of the complete elliptic integrals of the first and second kinds which are defined respectively by

$$K(m) = \int_0^{\pi/2} \left(1 - m \sin^2 t\right)^{-1/2} dt, \quad \text{for } 0 \le m < 1, \text{ and}$$

$$E(m) = \int_0^{\pi/2} \left(1 - m \sin^2 t\right)^{1/2} dt, \quad \text{for } 0 \le m \le 1.$$



B. Usage

B.1 Program Prototype

REAL YK, YE, SCPLTK, SCPLTE, EM

Assign a value to EM.

To compute the K() function:

YK =SCPLTK(EM)

To compute the E() function:

YE =SCPLTE(EM)

B.2 Argument Definitions

EM [in] Value of the parameter, m. Require $0 \le m < 1$ to compute K(m), and $0 \le m \le 1$ to compute E(m).

B.3 Modifications for Double Precision

For double precision usage, change the REAL statement to DOUBLE PRECISION and change the subroutine names from SCPLTK and SCPLTE to DCPLTK and DCPLTE.

C. Examples and Remarks

Example: Compute Legendre's relation

$$z = \pi/2 - (KE' + K'E - KK') = 0, \quad \text{where}$$
$$K = K(m), \quad K' = K(1-m)$$
$$E = E(m), \quad E' = E(1-m).$$

See DRSCPLTK and ODSCPLTK for an example of the use of these subprograms to evaluate this identity.

D. Functional Description

D.1 Properties of K and E

These functions are discussed in the references.

The function K(m) increases from $\pi/2$ to infinity as m varies from 0 to 1, and is asymptotic to $0.5 \ln(16/(1-m))$ as $m \to 1$. Although K(m) $\to \infty$ as $m \to 1$, the values of K(m) for computer representable values of m close to one are not extremely large. For example the value of K(m) at computer representable values of m is bounded by 10.6 on a machine having 10^{-8} precision and by 22.1 on a machine having 10^{-18} precision.

The function E(m) decreases from $\pi/2$ to 1 as m varies from 0 to 1.

The variable m used here is generally called the *parameter* of the elliptic functions. Other common parameterizations make use of the *modulus*, $k = \sqrt{m}$, or the *modular angle*, α, satisfying $k = \sin\ \alpha$.

D.2 Computation of K and E

These subprograms use Chebyshev polynomial approximators due to W. J. Cody, [3]. These are used in the form

$$P_n(1-m) - \ln(1-m) Q_n(1-m)$$

where P_n and Q_n are different polynomials for K and E, and n is the degree of the polynomials.

The negative logarithm base ten of the maximum absolute error of these approximators is given for degrees 5, 9, and 10 as follows:

Degree n	Precision of K approximator	Precision of E approximator
5	9.50	9.44
9	15.87	15.84
10	17.45	17.42

The subprograms use degree $n = 5$ on machines for which $-\log_{10}$(R1MACH(3)) < 8.2, degree $n = 9$ on machines for which $-\log_{10}$(R1MACH(3)) < 16.2, and degree $n = 10$ on machines having more precision. The accuracy of these subprograms is limited to 17.4 decimal places even on machines having more precision. See Chapter 19.1 for a description of R1MACH.

D.3 Accuracy Tests

Subprograms SCPLTK and SCPLTE were each tested on an IBM compatible PC using IEEE arithmetic by comparison with DCPLTK and DCPLTE, respectively, at 10,000 points in the interval (0.0, 1.0). The relative precision of the IEEE single precision arithmetic is $\rho = 2^{-23} \approx 1.192 \times 10^{-7}$.

For SCPLTK, 33% of the test points gave relative errors less than ρ. The maximum relative error observed was 2.0ρ.

For SCPLTE, 68% of the relative errors were less than ρ. The maximum relative error observed was 1.9ρ.

References

1. Milton Abramowitz and Irene A. Stegun, **Handbook of Mathematical Functions**, *Applied Mathematics Series 55*, National Bureau of Standards (1966) Chapter 17, 587–626.
2. J. F. Hart et al., **Computer Approximations**, J. Wiley and Sons, New York (1968) Section 6.9.
3. W. J. Cody, *Chebyshev approximations for the complete elliptic integrals K and E*, **Math. of Comp. 19**, 89–92 (1965) 105–112. See also [4].
4. W. J. Cody, *Corrigenda: "Chebyshev approximations for the complete elliptic integrals K and E"*, **Math. of Comp. 20**, 93–96 (1966) 207–207. See [3].

E. Error Procedures and Restrictions

The K subprograms issue an error message if $m < 0$ or $m \geq 1$. The E subprograms issue an error message if $m < 0$ or $m > 1$. On error conditions the value zero is returned. Error messages are processed using the subroutines of Chapter 19.2 with an error level of zero.

F. Supporting Information

The source language is ANSI Fortran 77.

Entry	Required Files
DCPLTE	AMACH, DCPLTE, DERM1, DERV1, ERFIN, ERMSG
DCPLTK	AMACH, DCPLTK, DERM1, DERV1, ERFIN, ERMSG
SCPLTE	AMACH, ERFIN, ERMSG, SCPLTE, SERM1, SERV1
SCPLTK	AMACH, ERFIN, ERMSG, SCPLTK, SERM1, SERV1

Designed and programmed by E. W. Ng, JPL, 1974. Modified by K. Stewart, JPL, 1981, C. L. Lawson and S. Y. Chiu, JPL, 1983.

DRSCPLTK

```
c     DRSCPLTK
c>> 2001-06-17 DRSCPLTK Krogh Changed T computation.
c>> 1996-05-30 DRSCPLTK Krogh Added external statement.
c>> 1994-10-19 DRSCPLTK Krogh  Changes to use M77CON
c>> 1994-09-01 DRSCPLTK WVS Moved formats to top for C conversion
c>> 1994-08-09 DRSCPLTK WVS set up for CHGTYP
c>> 1992-04-29 DRSCPLTK CAO Replaced '1' in format.
c>> 1991-11-19 DRSCPLTK CLL
c>> 1987-12-09 DRSCPLTK Lawson  Initial Code.
c--S replaces "?": DR?CPLTK, ?CPLTK, ?CPLTE
c
c     DEMONSTRATION DRIVER FOR ELLIPTIC INTEGRALS.
c
c     EVALUATE THE LEGENDRE'S RELATION:
c     Z = PI/2 - (K*E1 + K1*E - K*K1) = 0
c
      external R1MACH, SCPLTK, SCPLTE
      real             R1MACH, SCPLTK, SCPLTE
      real             EM(6),K,K1,E,E1, ONE
      real             PI2, T, TPRIME, Z, ZERO
      integer I
      data PI2 / 1.57079632679489661923132170E0 /
      data EM / 0.001E0, .2E0, .4E0, .6E0, .8E0, .999E0 /
      data ZERO,ONE / 0.E0, 1.E0 /
c
```

```
  200 format(5X,A2,9X,A10,7X,A10,8X,A1/' ')
  300 format(2X,F6.3,2X,F15.8,2X,F15.8,3X,G10.2)
  400 format(3X,A1,6X,F15.8,2X,F15.8,3X,G10.2)
  500 format(2X,F6.3,9X,A8,2X,F15.8,3X,G10.2)
  600 format(/' TPRIME = Machine epsilon =',E10.2)
  700 format(' T = 1. - TPRIME')
c
      TPRIME = R1MACH(4)
      T = ONE - TPRIME
c
      print 200,'EM','SCPLTK(EM)','SCPLTE(EM)','Z'
      print 300,ZERO,SCPLTK(ZERO),SCPLTE(ZERO)
c
      do 800 I = 1, 6
        K  = SCPLTK(EM(I))
        K1 = SCPLTK(1-EM(I))
        E  = SCPLTE(EM(I))
        E1 = SCPLTE(1-EM(I))
        Z  = PI2 - (K*E1 + K1*E - K*K1)
        print 300, EM(I), K, E, Z
  800 continue
      K  = SCPLTK( T )
      K1 = SCPLTK( TPRIME )
      E  = SCPLTE( T )
      E1 = SCPLTE( TPRIME )
      Z  = PI2 - (K*E1 + K1*E - K*K1)
      print 400, 'T', K, E, Z
      print 500, ONE, 'INFINITY', SCPLTE(ONE)
      print 600, TPRIME
      print 700
c
      end
```

ODSCPLTK

```
     EM          SCPLTK(EM)       SCPLTE(EM)         Z

   0.000         1.57079661       1.57079649
   0.001         1.57118952       1.57040381     -0.12E-06
   0.200         1.65962374       1.48903525     -0.12E-06
   0.400         1.77751958       1.39939225     -0.12E-06
   0.600         1.94956803       1.29842818     -0.12E-06
   0.800         2.25720549       1.17849004     -0.12E-06
   0.999         4.84113932       1.00217092     -0.12E-06
   T             9.35748768       1.00000060     -0.60E-06
   1.000           INFINITY       1.00000012

 TPRIME = Machine epsilon =  0.12E-06
 T = 1. - TPRIME
```